Lab 5

In this lab, you will learn how to visualize and do calculus on parametric curves and polar curves.

- Plot a parametric curve with `ParametricPlot`,
- Plot a polar curve using `PolarPlot`,
- Shade the region enclosed by curves using `ParametricPlot`,
- Find the enclosed area, arclength, and tangent lines.

1 To turn in

Do Problems 1-29 in a single Mathematica Notebook file, or ipynb file if you use JupyterLab. Write *your name* and *lab number* at the beginning of your report. Clearly label each problem to separate them from other problems. Make sure to comment on each problem. If your code doesn't run correctly, explain what you are trying to do. **Failed code without any comment/explanation will receive 0 point.** Submit on Canvas both the *pdf file* and the *source file* (nb or ipynb).

Problems	Points
1-5, 11, 12, 15, 18, 19, 24	1
9, 13, 14, 16, 17, 20-22, 27	2
6-8, 10, 23, 25, 26, 28, 29	3
Readability of your report	3
Total: 29	Total: 59

2 Plot a curve using ParametricPlot

The command **ParametricPlot** has several different functionalities, each of which has its own syntax. Only two of them will be mentioned here*. The first syntax is

```
ParametricPlot[{x,y}, {t,a,b}]
```

which draws a parametric curve $x = x(t)$, $y = y(t)$ with $t \in [a, b]$. The second syntax is

```
ParametricPlot[{x,y}, {t,a,b}, {s,c,d}]
```

which draws a *region* consisting of points (x, y) such that $x = x(t, s)$ and $y = y(t, s)$ with $t \in [a, b]$, $s \in [c, d]$. Drawing regions will be discussed in Section 4.

1. To draw the curve parametrized by $x = t \sin t$, $y = \cos t$, try the following

   ```
   ParametricPlot[{t*Sin[t], Cos[t]}, {t, -5, 5}]
   ```

   Change the color of the curve by putting an option `PlotStyle -> Red` inside the command.

2. Now change the range of t to $[-10, 10]$. Try with and without the option `AspectRatio->Automatic`. Report the difference.

3. An equivalent way to writing the command in Exercise 1 is:

*If you are interested in learning more functionalities of the command `ParametricPlot`, check out Wolfram's Documentation website: https://reference.wolfram.com/language/ref/ParametricPlot.html

```
x[t_] := t*Sin[t];
y[t_] := Cos[t];
ParametricPlot[{x[t], y[t]}, {t, -5, 5}]
```

What do you think an advantage of this way of writing is?

4. The parametric equation of a curve not only specifies the shape of the curve but also specifies how to draw the curve. You can create an interactive plot using **Manipulate** to see Mathematica draws this curve. This command works better on the desktop version of Mathematica than the cloud version. Try the following:

```
p[s_] :=  ParametricPlot[{t*Sin[t], Cos[t]}, {t, -5, s},
           PlotRange -> {{-5, 2}, {-1, 1}}]
Manipulate[p[s], {s, -4.9, 5}]
```

On the desktop version of Mathematica, you will see next to the plot a control bar with a Play button. If you click on it, you will see dynamic plotting as s moves from -4.9 to 5. Some dynamic plotting features are not available on the cloud version of Mathematica.

5. In the above command, the option `PlotRange` is to fix the plot window, which is $x \in [-5, 2]$ and $y \in [-1, 1]$ in this case. Remove the `PlotRange` option from the command, and run the command again. Allow s to run from -4.9 to 5 by clicking on the Play button or by dragging the cursor with your mouse. Report the difference.

6. Graph the curve parametrized by

$$
\begin{aligned}
x &= (e^{\cos t} - 2\cos(4t) - \sin^5(t/12))\sin t \\
y &= (e^{\cos t} - 2\cos(4t) - \sin^5(t/12))\cos t
\end{aligned}
$$

where $t \in [0, 20]$. Then use the command `Manipulate` to see how Mathematica plots this curve. Is it a closed and simple curve? Recall: "closed" means that the curve goes back to where it starts, and "simple" means that the curve doesn't intersect itself.

7. Same questions as Exercise 6 but with

$$
\begin{aligned}
x &= \sqrt{3}\cos(2t) - \cos(10t)\sin(20t) \\
y &= \sqrt{2}\sin(2t) - \sin(10t)\sin(20t)
\end{aligned}
$$

for $t \in [0, \pi]$.

8. A *cycloid* is the curve traced by a point on the rim of a circular wheel rolling along a straight line. Suppose the radius of the wheel is a. Then the parametric equation of the cycloid is

$$x = a(t - \sin t), \quad y = a(1 - \cos t)$$

Graph three cycloids with the wheel radii $a = 1$, $a = 1.5$, $a = 2$ on the same plot. You can plot each cycloid individually with a different color and combine them together using the command `Show`. You might need to specify a plot range in each command so that the combined plot is not cut off.

9. Plot the curve parametrized by $x = \cos t - \sin^2 t$, $y = \cos t \sin t$. Choose the range of t so that the curve is closed but doesn't loop more than once.

3 Plot a polar curve using PolarPlot

A polar curve is simply a curve given by an equation $r = r(\theta)$ in the polar coordinates. To make the notation more convenient, let us rename θ by t. The command **PolarPlot** can be used to draw a polar curve. To draw a polar curve $r = r(t)$, the syntax is

```
PolarPlot[r[t], {t,a,b}]
```

10. For example, to plot the polar curve $r = 1/t$ for $t \in [1, 50]$, try the following:

    ```
    PolarPlot[1/t, {t,1,50}, PlotRange -> Full]
    ```

    Can you guess if, as t increases, the spiral goes inside out or outside in? Explain your answer.

11. Use the command `Manipulate` to see how Mathematica draws this spiral.

12. The relation between polar coordinates (r, t) and Cartesian coordinates (x, y) is:

    $$x = r \cos t, \quad y = r \sin t$$

    Therefore, the curve in Exercise 10 can be described in Cartesian coordinates as

    $$x = \frac{1}{t} \cos t, \quad y = \frac{1}{t} \sin t$$

    Use the command `ParametricPlot` to draw the curve.

13. Plot the polar curve $r(t) = \sin(nt)$ with $n = 1, 2, 3, 4, 5, 6, 7$. For a general n, can you guess how many petals the flower has?

14. Plot the polar curve $r(t) = 1 + \cos(nt) + \sin^2(nt)$ with $n = 2, 3, 4, 5$. For which value of n do you get a shamrock leaf?

4 Shade the region enclosed by curves

The command

```
ParametricPlot[{x,y}, {t,a,b}, {s,c,d}]
```

to draw the *region* consisting of points (x, y) such that $x = x(t, s)$ and $y = y(t, s)$ with $t \in [a, b]$ and $s \in [c, d]$.

15. For example, any point lying under the parabola $y = x^2$, $0 \le x \le 2$, and above the x-axis has coordinates (x, y) where $0 \le x \le 2$, $0 \le y \le x^2$. Such a point (x, y) can be represented as

    $$x = t, \quad y = st^2$$

    where $0 \le t \le 2$ and $0 \le s \le 1$. Use the following command to sketch the region:

    ```
    ParametricPlot[{t, s t^2}, {t,0,2}, {s,0,1}]
    ```

16. As a good tip to keep in mind, if y is in between two numbers N and M, then you can write $y = N + s(M - N)$ for $0 \le s \le 1$. (In the previous exercise, $N = 0$ and $M = t^2$.) Use this rule and the command `ParametricPlot` to shade the region between two curves $y = \cos x$ and $y = \sin x$ for $0 \le x \le 8\pi$.

17. Draw the polar curve $r = 1 + \cos t$. This curve is called a *cardioid* (a heart shape).

18. To shade the region enclosed by this cardioid, note that every point inside the cardioid has polar coordinates (r, t) satisfying $0 \le t \le 2\pi$ and r in between 0 and $1 + \cos t$. Thus, $r = s(1 + \cos t)$ with $0 \le s \le 1$. Therefore, any point inside the curve has Cartesian coordinates
$$x = s(1 + \cos t)\cos t, \qquad y = s(1 + \cos t)\sin t$$
with $0 \le t \le 2\pi$ and $0 \le s \le 1$. To sketch this region, try the following command:

```
ParametricPlot[{s(1 + Cos[t])Cos[t], s(1 + Cos[t])Sin[t]},
                  {t, 0, 2 Pi}, {s, 0, 1}, PlotPoints -> 50]
```

19. Alternatively, you can break the command into

```
r[t_] := 1 + Cos[t];
ParametricPlot[{s*r[t]*Cos[t], s*r[t]*Sin[t]},
     {t, 0, 2 Pi}, {s, 0, 1}, PlotPoints -> 50]
```

This way of breaking down the command is helpful when r is a cumbersome expression of t, such as in the following exercise.

20. On the same plot, draw the cardioid $r = 1 + \cos t$ and shade the section inside the cardioid with $0 \le t \le \pi/2$. Hint: draw the cardioid and shade the section separately, and then combine them using the command `Show`.

21. Draw the polar curve
$$r = \frac{100}{100 + \left(t - \frac{\pi}{2}\right)^8}\left(2 - \sin(7t) - \frac{\cos(30t)}{2}\right)$$
for $t \in [-\frac{\pi}{2}, \frac{3\pi}{2}]$ and then shade the region enclosed by the curve using `ParametricPlot`.

22. On the same plot, draw the maple leaf in the previous exercise and shade the section inside the curve with $\pi/6 \le t \le 5\pi/6$.

23. Draw the polar curve
$$r = (1 - |t|)(1 + 3|t|), \quad t \in [-1, 1]$$
and then shade the region enclosed by the curve using `ParametricPlot`. Is the curve closed and simple?

5 Find enclosed area, arclength, tangent lines

Recall that the area and arclength of a parametric curve $x = x(t)$, $y = y(t)$, $a \le t \le b$, are given by
$$A = \left|\int_a^b xy'dt\right| = \left|\int_a^b yx'dt\right|, \quad L = \int_a^b \sqrt{(x')^2 + (y')^2}dt$$
The equation of the tangent line equation at the point $(x_0, y_0) = (x(t_0), y(t_0))$ is
$$v = y_0 + \frac{y'(t_0)}{x'(t_0)}(u - x_0) \tag{1}$$

If the denominator is equal to 0 then the tangent line is vertical: $u = x_0$. Here, the symbols u and v are used instead of x and y to avoid confusion in the coding.

Recall that the area and arclength of a polar curve $r = r(t)$, $a \le t \le b$ are given by

$$A = \int_a^b \frac{1}{2} r^2 dt, \quad L = \int_a^b \sqrt{(r')^2 + r^2} dt$$

Because the polar curve $r = r(t)$ is the same as the parametric curve $x = r(t)\cos t$, $y = r(t)\sin t$, you can find the equation of the tangent line using the formula (1).

24. Consider the parametric curve $x = t^2$, $y = t^3 - t$. To visualize the tangent line at the point corresponding to $t = 0.7$, you can use the following commands.

```
x[t_] := t^2;
y[t_] := t^3 - t;
t0 = 0.7;
p1 = ParametricPlot[{x[t], y[t]}, {t, -1.5, 1.5}];
p2 = Plot[y[t0]+y'[t0]/x'[t0](u - x[t0]), {u, 0, 2}, PlotStyle -> Red];
Show[p1, p2, PlotRange -> All]
```

25. Plot the curve $x = t^2 - 1$, $y = 2t^3 - t^2 - 2t + 2$. Then find the slopes of the tangent lines at the point where the curve intersect itself. Then draw these tangent lines together with the curve on the same plot.

26. Find the area of the "eye" (the loop) of the curve in Exercise 25.

27. Find the total length of the heart in Exercise 23 and the area enclosed by it.

28. Find the area and perimeter of the shaded region in Exercise 22. The perimeter consists of two straight edges and one curve.

29. Consider a four-petaled flower $r = \cos(2t)$. Shade one of the petals (of your choice). Then find the length and area of that petal.